LA-7879-MS
Informal Report

C.3

SNEX: Semianalytic Solution of the One-Dimensional Discrete Ordinates (S_n) Transport Equations with Diamond Differenced Angular Fluxes



LOS ALAMOS SCIENTIFIC LABORATORY
Post Office Box 1663 Los Alamos, New Mexico 87545

This report was not edited by the Technical Information staff.

UNITED STATES
DEPARTMENT OF ENERGY
CONTRACT W-7405-ENG. 36

LA-7879-MS
Informal Report
UC-32
Issued: June 1979

SNEX: Semianalytic Solution of the One-Dimensional Discrete Ordinates (S_n) Transport Equations with Diamond Differenced Angular Fluxes

B. R. Wienke

SNEX: SEMIANALYTIC SOLUTION OF THE ONE-DIMENSIONAL DISCRETE ORDINATES (S_n) TRANSPORT EQUATIONS WITH DIAMOND DIFFERENCED ANGULAR FLUXES

by

B. R. Wienke

ABSTRACT

SNEX is a code which exactly solves the spatial single group, one-dimensional (plane, cylinder, sphere) discrete ordinates transport equations with diamond approximation for the angular fluxes. Its purposes are to provide a standard for comparisons of spatial differencing schemes as well as an exact numerical solution to the discrete ordinates equations in one-dimensional geometries. Full solutions are generated by numerically integrating the inhomogeneous source terms and adding them to the homogeneous (analytic) solutions in the standard fashion. Simple relationships between angular quadratures permit application of the method to plane, cylindrical and spherical geometries. Analysis is confined to isotropic scattering. Simple theory and methodology are presented and discussed. A code listing and sample problems are also included.

I. INTRODUCTION

Typically, multigroup discrete ordinates codes solve the transport equations by using finite difference or finite element techniques on compatible meshes. As the meshes are refined, solutions converge to the exact values. An alternative, exact (but more time consuming) approach consists in solving the transport equations numerically by formal inversion of differential operators. Such procedure also has the advantage that exact solutions are generated independently of mesh size for given convergence criteria. In the following, we describe and detail SNEX, a transport equation solver which can be used to provide exact

numerical solutions to the linear transport equation in one dimensional plane, cylindrical and spherical geometrics.

SNEX solves the single group transport equation with isotropic scattering for specified left and right boundary conditions. It is structured closely to ONETRAN[1] as far as input/output formats. Coarse and fine meshes are defined with material zones specified on the coarse mesh. External sources of any functional form are admissible. Both the standard diamond and step starting schemes are options. Iteration on the scattering source is controlled by a convergence parameter computed from the difference in successive scalar flux estimates. To stabilize scattering iteration cycles, weighted averages of flux iterates may be employed. The code is structured for arbitrary S_n order, depending, of course, on available core storage (which scales roughly as the fine mesh size times the S_n order). All angular quadrature data, such as directions cosines, weights and angular coefficients, are read into the code directly. Output consists of pointwise angular and scalar fluxes computed on the fine mesh, as well as a listing of all input parameters and options. The solution algorithm requires numerical integration of an inhomogeneous source term (composed of external, scattering and angular edge components) and is effected with an adaptive Newton-Cotes (7-point) routine[2].

SNEX serves to provide an exact numerical solution to the monoenergetic discrete ordinates equations and a standard for comparison of differencing schemes.

II. TRANSPORT EQUATION, DISCRETE ORDINATES APPROXIMATION AND PHENOMENOLOGY

The time independent transport (linear) equation is written in the monoenergetic case,

$$\vec{\nabla}\cdot\vec{\Omega}\psi + \sigma\psi = \int d\Omega' \; \sigma_s \; \psi' + q \qquad (1)$$

where ψ is the particle flux (density times speed), σ,σ_s are the total, scattering cross sections, $\vec{\Omega},\vec{\Omega}'$ are unit vectors in the directions of particle travel and q is the external source of particles. The streaming operator, $\vec{\nabla}\cdot\vec{\Omega}$, takes the explicit form in the one dimensional geometries,

$$\vec{\nabla}\cdot\vec{\Omega}\ \psi = \mu \frac{\partial}{\partial x}\psi \qquad \text{(plane)}$$

$$= \frac{\mu}{\rho}\frac{\partial}{\partial \rho}(\rho\psi) - \frac{1}{\rho}\frac{\partial}{\partial \phi}(\eta\psi) \qquad \text{(cylinder)}$$

$$= \frac{\mu}{r^2}\frac{\partial}{\partial r}(r^2\psi) + \frac{1}{\rho}\frac{\partial}{\partial \mu}(1-\mu^2)\psi, \quad \text{(sphere)} \tag{2}$$

where μ is the angle between the position vector (in the three systems) and the direction of travel. The direction cosines satisfy,

$$\mu^2 + \eta^2 + \xi^2 = 1 \tag{3}$$

and,

$$\tan\phi = \frac{\eta}{\mu}\,. \tag{4}$$

The discrete ordinates approximation assume that the value of the angular flux, or ordinate, is determined at sets of discrete directions, $\Omega_m = (\mu_m, \eta_m, \xi_m)$ with m = 1, 2, 3, ... M. The corresponding angular flux is denoted generally by,

$$\psi_m(\vec{r}) = \psi(\vec{r},\Omega_m)\ , \tag{5}$$

and angular integrals and moments involving ψ_m are evaluated with quadrature weights, w_m, such that

$$\int \psi(\vec{r},\vec{\Omega})\,d\Omega = 4\pi \sum_{i=1}^{M} w_m\ \psi_m(\vec{r}) \tag{6}$$

for,

$$\sum_{i=1}^{M} w_m = 1 \tag{7}$$

Differencing of the angular variable is based upon the diamond approximation[3] between edge, $\psi_{m\pm1/2}$, and cell centered, ψ_m, fluxes,

$$2\psi_m = \psi_{m+1/2} + \psi_{m-1/2}, \tag{8}$$

which yields a simple recursion relationship for the edge (entering) flux in the m^{th} angular cell,

$$\psi_{m-1/2} = 2\psi_{m-1} - 2\psi_{m-2} + \cdots \mp 2\psi_1 \pm \psi_{1/2} \tag{9}$$

The angular differenced form of the transport equation in each of the three geometries is written,

$$\mu_m \frac{\partial \psi}{\partial x}m + \sigma\psi_m = S_m \qquad \text{(plane)}$$

$$\mu_m \frac{\partial}{\partial \rho}(\rho\psi_m) + (\alpha_{m+1/2}\ \psi_{m+1/2} - \alpha_{m-1/2}\ \psi_{m-1/2})w_m^{-1} + \rho\sigma\psi_m = \rho\ S_m \qquad \text{(cylinder)}$$

$$\mu_m \frac{\partial}{\partial r}(r^2\psi_m) + (\alpha_{m+1/2}\ \psi_{m+1/2} - \alpha_{m-1/2}\ \psi_{m-1/2})w_m^{-1}r + r^2\ \sigma\psi_m = r^2\ S_m \qquad \text{(sphere)} \tag{10}$$

with $\alpha_{m\pm1/2}$ sets of angular coefficients satisfying,

$$\alpha_{m+1/2} = \alpha_{m-1/2} = 0 \qquad \text{(plane)}$$

$$\alpha_{m+1/2} - \alpha_{m-1/2} = -w_m\ \mu_m \qquad \text{(cylinder)}$$

$$\alpha_{m+1/2} - \alpha_{m-1/2} = -2w_m\mu_m \qquad \text{(sphere)}$$

$$\alpha_{1/2} = \alpha_{M+1/2} = 0, \qquad \text{(all geometries)} \tag{11}$$

and S_m the sum of external and scattering sources,

$$S_m = q_m + \sum_{n=1}^{M} w_m \sigma_s (\mu_n \to \mu_m)\psi_n \cdot \tag{12}$$

Using Eqs. (8) and (11), we can rewrite Eqs. (10) into the simple general form,

$$\mu_m q \frac{\partial \psi}{\partial q}_m + \mu_m \gamma_m \psi_m + q \sigma \psi_m = q S_m + \mu_m \gamma_m \psi_{m-1/2} \tag{13}$$

where in the various geometries,

$q = x$

$\gamma_m = 0,$ (plane)

$q = \rho$

$\gamma_m = (\alpha_{m+1/2} + \alpha_{m-1/2})(w_m\mu_m)^{-1}$ (cylinder)

$q = r$

$\gamma_m = (\alpha_{m+1/2} + \alpha_{m-1/2})(w_m\mu_m)^{-1},$ (sphere) (14)

and the $\alpha_{m\pm1/2}$ still satisfy Eq. (11).

Equation (13) is easily inverted, with the result,

$$\psi_m = \psi_m^0 e^{\frac{-\sigma(q-q_0)}{\mu_m}} \left(\frac{q}{q_0}\right)^{-\gamma_m}$$

$$+ e^{\frac{-\sigma q}{\mu_m}} q^{-\gamma_m} \int_{q_0}^{q} dq' \, e^{\frac{\sigma q'}{\mu_m}} q'^{\gamma_m} \left[\frac{S_m}{\mu_m} + \frac{\gamma_m \psi_{m-1/2}}{q'} \right] , \qquad (15)$$

for ψ_m^0 the boundary angular flux at $q=q_0$. The content of Eq. (15) is straightforward. The first term on the right hand side is the homogeneous solution and the second term is the inhomogeneous integral. The pseudosource angular flux term, $\psi_{m-1/2}$, appearing in Eq. (15) is generated recusively from lower order solutions by means of Eq. (9). SNEX solves Eq. (15) by numerically integrating the source terms and adding them to the homogeneous solutions. If scattering is present, iteration takes place on the scattering term, Eq. (12), subject to satisfaction of the convergence criterion for the scalar flux.

The angular parameters α_m, w_m, μ_m are user specified. Typically in S_n applications, the quadrature sets $\{\mu_m, w_m\}$ are chosen to be Gaussian and the α_m are generated recursively from Eqs. (11) in the three geometries. The coefficients γ_m are automatically computed by SNEX.

III. METHODOLOGY

Equation (15) is evaluated numerically once boundary conditions and starting direction options are specified. Space/angle sweeps start at the right boundary for specified incoming boundary fluxes ($\mu_m < 0$) and proceed to the left boundary for specified incoming boundary fluxes ($\mu_m > 0$) and then return to the right boundary. Directions are angularly coupled through Eq. (9). In addition to boundary values, the value of $\psi_{1/2}$ must be specified to "start" the calculation. In cylindrical and spherical geometry, $\psi_{1/2}$ corresponds to the angular flux at $\mu = -1$, $\eta = 0$. Examination of Eqs. (2) reveals that $\psi_{1/2}$ then satisfies the slab-like equation in all three geometries,

$$-\frac{\partial \psi_{1/2}}{\partial q} + \sigma \psi_{1/2} = S_{1/2} \qquad (16)$$

which can be simply integrated for constant source $S_{1/2}$. Alternatively, one might make the step approximation,

$$\psi_{1/2} = \psi_1, \tag{17}$$

in which case the angular coupling is somewhat simplified in higher orders. Both options are available in SNEX. In the case of pure absorbing regions, the computation is completed after one space/angle sweep. If scattering is present, the scattering source, Eq. (12), is updated with the new angular flux iterate at the end of each space/angle sweep until convergence is met, or the iteration limit is exceeded.

The integral over the angular flux $\psi_{m+1/2}$ on the right hand side of Eq. (15) could be generated numerically from a corresponding lower order, $\ell \leq m-1$, equation of the same functional form. Thus, each angular flux term of order $\ell \leq m-1$, appearing on the right hand side of Eq. (15) would be the result of ℓ-dimensional nested integrations. Such procedure would be extremely time consuming and prone to divergence, particularly if negative fluxes occur either at cell edges or in the integrand. Early investigations[4] have demonstrated the occurrence of negative fluxes (due to the angular diamond approximation) for highly localized sources in absorbing regions. To avoid this situation, the effective source fluxes, $\psi_{m+1/2}$, appearing in Eq. (15) were approximated as piecewise linear functions on the fine mesh. That is, for given m in Eq. (15), each lower order flux term, $\ell \leq m-1$ computed earlier, is linearly interpolated across the mesh cell. As the mesh size decreases, the solutions converge to the exact result. For this reason, the overall approach in cylindrical and spherical geometries ($\gamma_m \neq 0$) is termed semianalytic.

IV. INPUT AND CODE PARAMETERS

Input to SNEX consists of the following parameters and arrays:

ANAME - title
A - S_n order (arbitrary)
ICT - coarse mesh intervals (arbitrary)
IACC - acceleration parameter (100-0)
IGEOM - geometry (1-plane, 2-cylinder, 3-sphere)
IBL - left boundary condition (0-specified flux, 1-reflective)

IBR - right boundary condition (0-specified flux)

IL - iteration limit (arbitrary)

ISTEP - starting direction option (0-slab exact, 1-step)

CS(A) - direction cosines

W(A) - weights

B(A+1) - angular coefficients

X(ICT+1) - coarse mesh

IIT(ICT) - fine mesh intervals/coarse mesh

SG(ICT) - total cross sections/coarse mesh

ST(ICT) - scattering cross sections/coarse mesh

Q(ICT) - external source/coarse mesh

FB(1,A/2) - left boundary flux,

FB(IFT+1,A/2) - right boundary flux

with IFT the number of fine mesh points. Additionally the following parameters and arrays are employed in SNEX:

Z(IFT+1) - fine mesh points

DX(ICT) - fine mesh interval widths/coarse mesh

FB(IFT+1,A+1) - angular flux

SF(IFT+1) - scalar flux

PB(IFT+1,A+1) - previous iterate angular flux

CV - convergence parameter

IN - iteration count

RIACC - acceleration parameter (IACC/100).

In present form, A+1, IFT+1 are set at 17 and 301 in array statements, while ICT is set to 10. The user is free to change these parameters to suit his needs. The left boundary flux FB(1,A/2) for outgoing directions is only read if IGEOM = 1 (plane). Otherwise, in curved (cylinder, sphere) geometries (IGEOM = 2,3), a reflective boundary condition is assigned at the origin (left boundary). The acceleration parameter IACC is used to fractionally weight presently computed values of the angular flux with the previous flux iterates. The weighting parameter used on the k^{th} flux iterate is RIACC = IACC/100 while the

parameter RIACCM = 1.-RIACC is used on the $(k-1)^{th}$ flux iterate. In most applications, successive flux iterations monotonically approach the convergence criterion and one uses IACC = 100. In oscillatory situations, further mesh refinement and/or taking IACC<100 will aid convergence. If ISTEP=0 is chosen as the starting direction option, $\psi_1 = \psi_{1/2}$, yielding a slab solution for ψ_1, as seen from Eq. (13). If ISTEP = 1, $\psi_1 \neq \psi_{1/2}$, and the slab equation for $\psi_{1/2}$ provides the starting point for space/angle sweeps.

The origin is a singular point in curved geometries, as seen from Eq. (13) or (15). Integrations over the origin diverge (overflow). In these cases, the origin is automatically set to $1x10^{-6}$ for computational simplicity.

The data are read into SNEX in the following order and formats:

ANAME(10)	10A8
A,ICT,IACC,IGEOM,IBL,IBR,ISTEP	10(I7,1x)
CS(A)	5(E14.8,2x)
W(A)	5(E14.8,2x)
B(A)	5(E14.8,2x)
X(ICT)	5(E14.8,2x)
IIT(ICT)	10(I7,1x)
SG(ICT)	10(F7.2,1x)
ST(ICT)	10(F7.2,1x)
Q(ICT)	10(F7.2,1x)
FB(1,A/2)	5(E14.8,2x)
FB(IFTP,A/2)	5(E14.8,2x).

V. SAMPLE PROBLEM

As a sample problem, we consider a scattering/absorbing sphere (IGEOM=3) 1 mean free path (mfp) thick with uniform $\sigma_s/\sigma_t = .5$. An isotropic source is placed in a localized slice of thickness .00002 mfp at unit distance from the origin. Reflective boundary conditions are automatically assigned at the left boundary and the incoming right boundary flux is taken to be zero. An S_4 angular quadrature (Gaussian) is assigned and the slab starting direction option is used (ISTOP=0). The acceleration parameter is set to maximum value (IACC=100) and the iteration limit to 30 (IL=30). The sphere is coarse meshed into 3 regions of approximately .1 mfp (fine mesh) intervals.

The input file for the above sample problem is given below.

```
     ABSORBER/SCATTERER SOURCE TEST PROBLEM
       4        3      100        3        1        0       30       0
 -.86113630E+00  -.33998100E+00   .33998100E+00   .86113630E+00
  .17392700E+00   .32607200E+00   .32607200E+00   .17392700E+00
  .00000000E+00   .29955000E+00   .52126700E+00   .29955000E+00  .00000000E+00
  .10000000E-05   .99000000E-02   .10100000E-01   .10000000E+02
       1       1       10
     .10     .10     .10
    0.05    0.05    0.05
    0.00    1.00    0.00
    0.00    1.00    0.00
    0.00    1.00    0.00
    0.00    1.00    0.00
  .00000000E+00   .00000000E+00
```

The corresponding output is listed below. Entries are self-explanatory. Angular and scalar fluxes are given. For a given order A, there are A+1 angular fluxes printed, with the $(A+1)^{st}$ entry the starting direction flux, $\psi_{1/2}$. The last two (unlabeled), numbers are the iteration count (IN) and acceleration parameter (RIACC).

```
     ABSORBER/SCATTERER SOURCE TEST PROBLEM


       4 ORDER
       3 INTERVALS
     100 ACCELERATION
       3 PLANE/CYLINDER/SPHERE
       1 LEFT BDY/SPECIFIED/REFLECTIVE
       0 RIGHT BDY/SPECIFIED
      30 ITERATION LIMIT
       0 STARTING/STEP


     COARSE MESH

 .10000000E-05   .99000000E-02   .10100000E-01   .10000000E+02
     FINE MESH

       1       1       10
 .10000000E-05   .99000000E-02   .10100000E-01   .10090900E+01   .20080800E+01
 .30070700E+01   .40060600E+01   .50050500E+01   .60040400E+01   .70030300E+01
 .80020200E+01   .90010100E+01   .10000000E+02
     CROSS SECTIONS

    .10     .10     .10
    .05     .05     .05
     SOURCE

   0.00    1.00    0.00
   0.00    1.00    0.00
   0.00    1.00    0.00
   0.00    1.00    0.00
     QUADRATURE(GAMMA/BETA/COSINE/WEIGHT)

 .17222743E+01   .25172876E+01   .25172876E+01   .17222743E+01
0.               .29955000E+00   .52126700E+00   .29955000E+00  0.
-.86113630E+00  -.33998100E+00   .33998100E+00   .86113630E+00
 .17392700E+00   .32607200E+00   .32607200E+00   .17392700E+00
```

```
     LEFT/RIGHT BOUNDARY FLUXES

0.                0.

  POSITION        ANGULAR FLUX
  .100000E-05     .224181E-03   .224182E-03   .224182E-03   .224181E-03
                  .224181E-03
  .990000E-02     .255937E-03   .589812E-03   .812321E-03   .563086E-03
                  .224199E-03
  .101000E-01     .243263E-04   .245563E-04   .130919E-02   .834020E-03
                  .241952E-04
  .100909E+01     .782870E-05   .104182E-04   .166968E-04   .842342E-03
                  .709065E-05
  .200808E+01     .284647E-05   .381839E-05   .799981E-05   .203276E-03
                  .262568E-05
  .300707E+01     .150857E-05   .204510E-05   .407675E-05   .861263E-04
                  .139514E-05
  .400606E+01     .911333E-06   .125765E-05   .252053E-05   .456267E-04
                  .841656E-06
  .500505E+01     .581500E-06   .824681E-06   .169865E-05   .273466E-04
                  .534350E-06
  .600404E+01     .375995E-06   .555322E-06   .119909E-05   .177189E-04
                  .342760E-06
  .700303E+01     .235725E-06   .372099E-06   .867638E-06   .121113E-04

                  .213554E-06
  .800202E+01     .136243E-06   .235955E-06   .631622E-06   .860545E-05
                  .121278E-06
  .900101E+01     .599907E-07   .120848E-06   .449721E-06   .629564E-05
                  .525233E-07
  .100000E+02     0.            0.            .285307E-06   .471160E-05
                  0.

  POSITION        SCALAR FLUX
  .10000000E-05   .22418071E-03
  .99000000E-02   .59964694E-03
  .10100000E-01   .58418792E-03
  .10090900E+01   .15570912E-03
  .20080800E+01   .39703920E-04
  .30070700E+01   .17238235E-04
  .40060600E+01   .93261853E-05
  .50050500E+01   .56802532E-05
  .60040400E+01   .37192616E-05
  .70030300E+01   .25518983E-05
  .80020200E+01   .18033086E-05
  .90010100E+01   .12911359E-05
  .10000000E+02   .91250462E-06

       9    1.000000
```

VI. CODE LISTING

The present code listing of SNEX used on the CDC 7600 under the Livermore Time Sharing System (LTSS) is listed below. All reads and writes to unit 59 are commands to a terminal. Units 5 and 6 are the standard input and output devices.

LASL Identification No. LP-1066.

```
      PROGRAM SNEX (INPUT,OUTPUT,TTY,TAPE5=INPUT,TAPE6=OUTPUT,TAPE59=TTY
     1 )
      COMMON /QNC7OUT/ HOWFA7, VALU7
      COMMON Z(301), SG(10), FB(301,17), CS(16), G(16), SR(300,16), ST(1
     1 0)
      COMMON PB(301,17), W(16), K, M, KC, AH, ISTEP
      DIMENSION SF(301), ANAME(10)
      DIMENSION X(11), DX(10), IIT(10), Q(10,16), B(17), KG(300)
      INTEGER A, AP, AH, AHP
C           READ DATA
      READ (5,280) (ANAME(I),I=1,10)
      CV=1.E-4
      TMAX=1.0
      READ (5,290) A,ICT,IACC,IGEOM,IBL,IBR,IL,ISTEP
      AH=A/2
      AP=A+1
      AHP=AH+1
      RIACC=IACC
      RIACC=RIACC/100.
      RIACCM=1.-RIACC
      READ (5,390) (CS(I),I=1,A)
      READ (5,390) (W(I),I=1,A)
      READ (5,390) (B(I),I=1,AP)
      ICTP=ICT+1
      READ (5,390) (X(I),I=1,ICTP)
      IF (X(1).EQ.0.00.AND.IGEOM.NE.1) X(1)=1.E-6
      READ (5,290) (IIT(I),I=1,ICT)
      READ (5,270) (SG(I),I=1,ICT)
      READ (5,270) (ST(I),I=1,ICT)
      DO 10 N=1,A
   10 READ (5,270) (Q(I,N),I=1,ICT)
C           FORM MESH.MATERIAL REGIONS
      IFT=0
      DO 20 I=1,A
   20 G(I)=(B(I)+B(I+1))/W(I)
      DO 30 I=1,ICT
      FT=IIT(I)
      IFT=IFT+IIT(I)
   30 DX(I)=(X(I+1)-X(I))/FT
      IFTP=IFT+1
      READ (5,390) (FB(IFTP,I),I=1,AH)
      FB(IFTP,AP)=FB(IFTP,1)
      IF (IGEOM.EQ.2.OR.IGEOM.EQ.3) IBL=1
      IF (IBL.EQ.1) GO TO 40
      READ (5,390) (FB(1,I),I=AHP,A)
      FB(1,AP)=FB(1,1)
   40 CONTINUE
      DO 50 I=1,IFTP
      DO 50 J=1,A
   50 PB(I,J)=0.00
      L=0
      DO 70 I=1,ICT
      IT=IIT(I)
      DO 70 J=1,IT
      L=L+1
      DO 60 N=1,A
   60 SR(L,N)=Q(I,N)
      KG(L)=I
   70 Z(L)=X(I)+(J-1)*DX(I)
      Z(IFTP)=X(ICTP)
C           WRITE DATA
      WRITE (6,290) (ANAME(I),I=1,10)
      WRITE (6,410)
```

```
      WRITE (6,380) A,ICT,IACC,IGEOM,IBL,IBR,IL,ISTEP
      WRITE (6,300)
      WRITE (6,390) (X(I),I=1,ICTP)
      WRITE (6,310)
      WRITE (6,290) (IIT(I),I=1,ICT)
      WRITE (6,390) (Z(I),I=1,IFTP)
      WRITE (6,320)
      WRITE (6,270) (SG(I),I=1,ICT)
      WRITE (6,270) (ST(I),I=1,ICT)
      WRITE (6,330)
      DO 80 N=1,A
   80 WRITE (6,270) (Q(I,N),I=1,ICT)
      WRITE (6,340)
      WRITE (6,390) (G(I),I=1,A)
      WRITE (6,390) (B(I),I=1,AP)
      WRITE (6,390) (CS(I),I=1,A)
      WRITE (6,390) (W(I),I=1,A)
      WRITE (6,370)
      IF (IBL.EQ.0) WRITE (6,390) (FB(1,I),I=AHP,A)
      WRITE (6,390) (FB(IFTP,I),I=1,AH)
      WRITE (59,280) (ANAME(I),I=1,10)
      WRITE (59,410)
      WRITE (59,380) A,ICT,IACC,IGEOM,IBL,IBR,IL,ISTEP
      WRITE (59,300)
      WRITE (59,390) (X(I),I=1,ICTP)
      WRITE (59,310)
      WRITE (59,290) (IIT(I),I=1,ICT)
      WRITE (59,390) (Z(I),I=1,IFTP)
      WRITE (59,320)
      WRITE (59,270) (SG(I),I=1,ICT)
      WRITE (59,270) (ST(I),I=1,ICT)
      CH=0.00
      DO 90 I=1,ICT
   90 IF (ST(I).NE.0.00) CH=ST(I)
      WRITE (59,330)
      DO 100 N=1,A
  100 WRITE (59,270) (Q(I,N),I=1,ICT)
      WRITE (59,340)
      WRITE (59,390) (G(I),I=1,A)
      WRITE (59,390) (B(I),I=1,AP)
      WRITE (59,390) (CS(I),I=1,A)
      WRITE (59,390) (W(I),I=1,A)
C          SPACE/ANGLE SWEEPS
      IN=0
  110 CONTINUE
      IN=IN+1
      WRITE (59,420) IN,TMAX
      KC=1
      DO 120 I=1,IFT
      K=IFT-I+1
      Y=Z(K)
      M=KG(K)
      FB(K,AP)=S1(Y)
  120 CONTINUE
      DO 140 I=1,IFT
      K=IFT-I+1
      Y=Z(K)
      M=KG(K)
      DO 130 J=1,AH
      KC=J
      FB(K,J)=FF(Y)
      IF (J.EQ.1.AND.ISTEP.EQ.1) FB(K,1)=S1(Y)
  130 CONTINUE
```

```
  140 CONTINUE
      IF (IBL.NE.1) GO TO 160
      DO 150 I=AHP,A
  150 FB(1,I)=FB(1,A-I+1)
  160 CONTINUE
      DO 180 I=1,IFT
      K=I
      Y=Z(K+1)
      M=KG(K)
      DO 170 J=AHP,A
      KC=J
  170 FB(K+1,J)=FF(Y)
  180 CONTINUE
      IF (CH.EQ.0.00) GO TO 220
C          CONVERGENCE TEST
      TMAX=CV
      DO 200 I=1,IFTP
      TP=0.0
      TF=0.0
      DO 190 J=1,A
      TP=TP+PB(I,J)*W(J)
      TF=TF+FB(I,J)*W(J)
  190 CONTINUE
      CMAX=ABS((TP-TF)/TF)
      IF (CMAX.GT.TMAX) TMAX=CMAX
  200 CONTINUE
      IF (TMAX.LE.CV) GO TO 220
      DO 210 I=1,IFTP
      DO 210 J=1,A
  210 PB(I,J)=FB(I,J)*RIACC+PB(I,J)*RIACCM
      IF (IN.GT.IL) GO TO 220
      GO TO 110
  220 CONTINUE
C          WRITE OUTPUT
      WRITE (59,410)
      DO 240 I=1,IFTP
      SF(I)=0.0
      DO 230 J=1,A
  230 SF(I)=SF(I)+FB(I,J)*W(J)
  240 CONTINUE
      WRITE (6,350)
      DO 250 I=1,IFTP
  250 WRITE (6,400) (Z(I),(FB(I,K),K=1,AP))
      WRITE (6,360)
      DO 260 I=1,IFTP
  260 WRITE (6,390) (Z(I),SF(I))
      WRITE (6,410)
      WRITE (6,420) IN,RIACC
      STOP
C
  270 FORMAT (10(F7.2,1X))
  280 FORMAT (10A8)
  290 FORMAT (10(I7,1X))
  300 FORMAT (1X,4X,11HCOARSE MESH/)
  310 FORMAT (1X,4X,9HFINE MESH/)
  320 FORMAT (1X,4X,14HCROSS SECTIONS/)
  330 FORMAT (1X,4X,6HSOURCE/)
  340 FORMAT (1X,4X,36HQUADRATURE(GAMMA/BETA/COSINE/WEIGHT)/)
  350 FORMAT (//2X,8HPOSITION,7X,12HANGULAR FLUX)
  360 FORMAT (/1X,8HPOSITION,8X,11HSCALAR FLUX)
  370 FORMAT (1X,4X,26HLEFT/RIGHT BOUNDARY FLUXES/)
  380 FORMAT (1X,4X,I3,6H ORDER/5X,I3,10H INTERVALS/5X,I3,13H ACCELERATI
     1ON/5X,I3,22H PLANE/CYLINDER/SPHERE/5X,I3,30H LEFT BDY/SPECIFIED/RE
```

```
     2FLECTIVE/5X,I3,20H RIGHT BDY/SPECIFIED/5X,I3,16H ITERATION LIMIT/5
     3 X,I3,14H STARTING/STEP//)
  390 FORMAT (5(E14.8,2X))
  400 FORMAT (1X,E12.6,3X,4(E12.6,1X)/16X,4(E12.6,1X)/16X,4(E12.6,1X)/16
     1 X,5(E12.6,1X))
  410 FORMAT (/)
  420 FORMAT (1X,I6,2X,F10.6)
      END
      FUNCTION QNC7 (FOF,Y1,Y2,FERR,MXEVALS,KOUN7)
C           ADAPTIVE NEWTON-COTES INTEGRATION ROUTINE
      INTEGER DELTEV
      COMMON /QNC7OUT/ HOWFA7, VALU7
      DIMENSION FOT(20), F1T(20), F2T(20), F3T(20), F4T(20), F5T(20),
     1 F6T(20), F(13), W(4), LEG(20), DXT(20), XMT(20), ART(20), EPST(20
     2 ), ESTT(20), SUM1(20)
      DATA W /4.8809523809523809524E-02,2.5714285714285714285E-01,3.214
     1 2857142857142857E-02,3.2380952380952380952E-01/
      DATA TWLTH /0.083333333333333333/
      KINDCR=+1
      MAXEV=MXEVALS
      DELTEV=MXEVALS
      LMAX=20
      A=Y1
      B=Y2
      EPS=FERR
      DA=B-A
      AREA=1.
      EST=1.
      L=0
      H2=DA*TWLTH
      DO 10 I=1,13,2
      PT=A+H2*(I-1)
      F(I)=FOF(PT)
   10 CONTINUE
      KOUN7=7
   20 DX=0.5*DA
      H2=DA*TWLTH
      XM=A+DX
      DO 30 I=2,12,2
      PT=A+H2*(I-1)
      F(I)=FOF(PT)
   30 CONTINUE
      QNC7=0.
      KOUN7=KOUN7+6
      IF (KOUN7.GT.MAXEV) GO TO 100
   40 ESTL=(W(1)*(F(1)+F(7))+W(2)*(F(2)+F(6))+W(3)*(F(3)+F(5))+W(4)*F(4)
     1 )*DX
      ESTR=(W(1)*(F(7)+F(13))+W(2)*(F(8)+F(12))+W(3)*(F(9)+F(11))+W(4)*F
     1 (10))*DX
      AREA=AREA-ABS(EST)+ABS(ESTL)+ABS(ESTR)
      SUM=ESTL+ESTR
      IF (ABS(EST-SUM)-EPS*AREA) 50,50,60
   50 IF (EST-1.) 80,60,80
   60 IF (L-LMAX) 70,80,80
   70 L=L+1
      LEG(L)=2
      FOT(L)=F(7)
      F1T(L)=F(8)
      F2T(L)=F(9)
      F3T(L)=F(10)
      F4T(L)=F(11)
      F5T(L)=F(12)
      F6T(L)=F(13)
```

```
      DXT(L)=DX
      XMT(L)=XM
      ART(L)=AREA
      EPST(L)=EPS/1.4
      ESTT(L)=ESTR
      DA=DX
      F(13)=F(7)
      F(11)=F(6)
      F(9)=F(5)
      F(7)=F(4)
      F(5)=F(3)
      F(3)=F(2)
      EST=ESTL
      EPS=EPST(L)
      GO TO 20
   80 IF (LEG(L)-2) 140,90,90
   90 SUM1(L)=SUM
      LEG(L)=1
      A=XMT(L)
      DA=DXT(L)
      F(1)=F0T(L)
      F(3)=F1T(L)
      F(5)=F2T(L)
      F(7)=F3T(L)
      F(9)=F4T(L)
      F(11)=F5T(L)
      F(13)=F6T(L)
      AREA=ART(L)
      EST=ESTT(L)
      EPS=EPST(L)
      GO TO 20
  100 CONTINUE
      IF (KINDCR.EQ.+1) GO TO 120
  110 DELTEV=0.5*DELTEV
      MAXEV=MAXEV+DELTEV
      LMAX=10.*(Y2-A)/(Y2-Y1)+1.
      KINDCR=-1
      GO TO 40
  120 HOWFA7=A
      VALU7=0.0
      IL=L
  130 IF (IL.LT.1) GO TO 110
      IF (LEG(IL).EQ.1) VALU7=VALU7+SUM1(IL)
      IL=IL-1
      GO TO 130
  140 SUM=SUM1(L)+SUM
      L=L-1
      IF (L.LT.1) GO TO 150
      GO TO 80
  150 QNC7=SUM
      KOUN7=ISIGN(KOUN7,KINDCR)
      RETURN
      END
      FUNCTION R1 (Y)
C           SLAB SOLUTION INTEGRAND
      COMMON Z(301), SG(10), FB(301,17), CS(16), G(16), SR(300,16), ST(1
     1 0)
      COMMON PB(301,17), W(16), K, M, KC, AH, ISTEP
      EXTERNAL SFUN, SSS
      INTEGER AH
      YB=Z(K+1)
      SB=SG(M)
      CB=-1.0
```

```
      IF (ISTEP.EQ.1) CB=CS(1)
      SS=SFUN(Y)+SSS(Y)
      R1=EXP(SB*Y/CB)*SS/CB
      RETURN
      END
      FUNCTION S1 (Y)
C          SLAB SOLUTION STARTING DIRECTION
      COMMON Z(301), SG(10), FB(301,17), CS(16), G(16), SR(300,16), ST(1
     1 0)
      COMMON PB(301,17), W(16), K, M, KC, AH, ISTEP
      EXTERNAL QNC7, R1
      INTEGER AH
      KP=2*AH+1
      YB=Z(K+1)
      SB=SG(M)
      CB=-1.0
      IF (ISTEP.EQ.1) CB=CS(1)
      F=FB(K+1,KP)
      S1=F*EXP(-SB*(Y-YB)/CB)+EXP(-SB*Y/CB)*QNC7(R1,Y,YB,1.E-4,500,KOUN7
     1 )*(-1.)
      RETURN
      END
      FUNCTION QQ (Y)
C          INHOMOGENEOUS TOTAL SOURCE INTEGRAND
      COMMON Z(301), SG(10), FB(301,17), CS(16), G(16), SR(300,16), ST(1
     1 0)
      COMMON PB(301,17), W(16), K, M, KC, AH, ISTEP
      EXTERNAL RR, SSS, SFUN
      INTEGER AH
      SB=SG(M)
      CB=CS(KC)
      GB=G(KC)
      SS=SFUN(Y)+SSS(Y)
      QQ=Y**(GB/CB)*EXP(SB*Y/CB)*(SS/CB+GB/(CB*Y)*RR(Y))
      RETURN
      END
      FUNCTION FF (Y)
C          INHOMOGENEOUS TOTAL SOURCE INTEGRAL
      COMMON Z(301), SG(10), FB(301,17), CS(16), G(16), SR(300,16), ST(1
     1 0)
      COMMON PB(301,17), W(16), K, M, KC, AH, ISTEP
      EXTERNAL QQ, QNC7
      INTEGER AH
      CB=CS(KC)
      GB=G(KC)
      SB=SG(M)
      IF (KC.GT.AH) GO TO 10
      F=FB(K+1,KC)
      YB=Z(K+1)
      FF=F*EXP(-SB*(Y-YB)/CB)*(YB/Y)**(GB/CB)+EXP(-SB*Y/CB)*Y**(-GB/CB)
     1 *QNC7(QQ,Y,YB,1.E-4,500,KOUN7)*(-1.)
      GO TO 20
   10 F=FB(K,KC)
      YB=Z(K)
      FF=F*EXP(-SB*(Y-YB)/CB)*(YB/Y)**(GB/CB)+EXP(-SB*Y/CB)*Y**(-GB/CB)
     1 *QNC7(QQ,YB,Y,1.E-4,500,KOUN7)
   20 CONTINUE
      RETURN
      END
      FUNCTION RR (Y)
C          LINEAR INTERPOLATION OF PSEUDOSOURCE EDGE FLUX
      COMMON Z(301), SG(10), FB(301,17), CS(16), G(16), SR(300,16), ST(1
     1 0)
```

```
      COMMON PB(301,17), W(16), K, M, KC, AH, ISTEP
      INTEGER AH
      DIMENSION R(17)
      DO 20 KK=1,KC
      KT=KK-1
      IF (KK.EQ.1) KT=2*AH+1
      IF (KC.GT.AH) GO TO 10
      YB=Z(K+1)
      R(KK)=(PB(K,KT)-PB(K+1,KT))/(Z(K)-Z(K+1))*(Y-YB)+PB(K+1,KT)
      GO TO 20
   10 YB=Z(K)
      R(KK)=(PB(K+1,KT)-PB(K,KT))/(Z(K+1)-Z(K))*(Y-YB)+PB(K,KT)
   20 CONTINUE
      TOT=0.00
      IC=1
      DO 30 I=1,KC
      IC=IC+1
      J=KC-I+1
      WT=2.0
      IF (J.EQ.1) WT=1.0
      SN=(-1)**IC
   30 TOT=TOT+SN*WT*R(J)
      RR=TOT
      RETURN
      END
      FUNCTION SSS (Y)
C          SCATTERING SOURCE FUNCTION
      COMMON Z(301), SG(10), PB(301,17), CS(16), G(16), SR(300,16), ST(1
     1 0)
      COMMON PB(301,17), W(16), K, M, KC, AH, ISTEP
      INTEGER AH
      IF (ST(M).EQ.0.00) RETURN
      A=0.00
      NN=2*AH
      DO 20 J=1,NN
      IF (KC.GT.AH) GO TO 10
      YB=Z(K+1)
      AB=(PB(K,J)-PB(K+1,J))/(Z(K)-Z(K+1))*(Y-YB)+PB(K+1,J)
      A=A+ST(M)*W(J)*AB
      GO TO 20
   10 YB=Z(K)
      AB=(PB(K+1,J)-PB(K,J))/(Z(K+1)-Z(K))*(Y-YB)+PB(K,J)
      A=A+ST(M)*W(J)*AB
   20 CONTINUE
      SSS=A
      RETURN
      END
      FUNCTION SFUN (Y)
C          EXTERNAL SOURCE FUNCTION
      COMMON Z(301), SG(10), PB(301,17), CS(16), G(16), SR(300,16), ST(1
     1 0)
      COMMON PB(301,17), W(16), K, M, KC, AH, ISTEP
      INTEGER AH
      SFUN=SR(K,KC)
      RETURN
      END
```

REFERENCES

1. T. R. Hill, "ONETRAN, A Discrete Ordinates Finite Element Code For The Solution Of The One-Dimensional Multigroup Transport-Equation, LASL report LA-5990-MS (June 1975).

2. D. Kahaner, "Comparison of Numerical Quadrature Formulas," LASL report LA-4137 (June 1969).

3. B. G. Carlson, "Solution of the Transport Equation by S_n Approximations," USAEC report LA-1599 (March 1953)

4. B. R. Wienke "Transport and Reactor Theory, July 1 - September 30, 1978," W. F. Miller, Jr. (ed), LASL report LA-7554-PR (November 1978).

5. R. E. Alcouffe, E. W. Larsen, W. F. Miller, Jr., B. R. Wienke, "Calculational Efficiency of Numerical Transport Methods: Slab Geometry" Trans. Am. Nuc. Soc. 30, 250 (1978).

APPENDIX A.

DIFFERENCE SCHEMES AND INVERSION OF THE ONE DIMENSIONAL TRANSPORT EQUATION

As described in the foregoing, SNEX formally inverts the discrete ordinates equations with diamond approximation in angular variable by numerically integrating the inhomogeneous term. Partitioning the spatial domain into i = 1, 2, 3, ... I, with cell boundaries at $q_{i\pm1/2}$

$$q_{i-1/2} \leq q_i \leq q_{i+1/2}$$

$$\Delta = q_{i+1/2} - q_{i-1/2} \qquad \text{(A-1)}$$

in the i^{th} cell, the general solution can be written in the standard representation

$$\psi_{i,m}(q) = \psi_{i-1/2,m}\, e^{\frac{-\sigma(q-q_{i-1/2})}{\mu_m}} \left(\frac{q}{q_{i-1/2}}\right)^{-\gamma_m}$$

$$+ e^{\frac{-\sigma q}{\mu_m}} q^{-\gamma_m} \int_{q_{i-1/2}}^{q} dq' \, e^{\frac{rq'}{\mu_m}} q'^{\gamma_m} \left[\frac{S_{i,m}}{\mu_m} + \frac{\gamma_m \psi_{i,m-1/2}}{q'} \right] \qquad \text{(A-2)}$$

with i denoting the cell and i-1/2 the cell edge (left). The form of Eq. (A-2) suggests bases for differencing schemes.

In plane geometry, $\gamma_m = 0$, and Eq. (A-1) reduces to

$$\psi_{i,m} = \psi_{i-1/2,m} e^{\frac{-\sigma(x-x_{i-1/2})}{\mu_m}} + e^{\frac{-\sigma x}{\mu_m}} \int_{x_{i-1/2}}^{x} dx' \, e^{\frac{\sigma x'}{\mu_m}} \frac{S_{i,m}}{\mu_m} \qquad \text{(A-3)}$$

For $S_{i,m}$ a polynomial in x', the integrand in Eq. (A-3) can always be directly inverted. Cases for $S_{i,m}$ a constant and linear function from the basis for the recently developed and investigated characteristic method[5] in slabs, an extremely accurate and efficient scheme.

Angular coupling ($\gamma_m \neq 0$) in curvilinear geometries modifies the basic exponential scheme in plane geometry by powers of q^{γ_m}. In cylindrical and spherical geometries, a basic difference scheme of the form,

$$\psi_{i,m}(q) = \psi_{i-1/2,m} e^{\frac{-\sigma(q-q_{i-1/2})}{\mu_m}} \left(\frac{q}{q_{i-1/2}} \right)^{-\gamma_m} + T_{i,m}(q_{i-1/2}, q) \qquad \text{(A-4)}$$

is advocated, where $T_{i,m}$ represents the inhomogeneous source,

$$T_{i,m}(q_{i-1/2}, q) = e^{\frac{-\sigma q}{\mu_m}} q^{-\gamma_m} \int_{q_{i-1/2}}^{q} dq' \, e^{\frac{\sigma q'}{\mu_m}} q'^{\gamma_m}$$

$$\left[\frac{S_{i,m}}{\mu_m} + \frac{\gamma_m \psi_{i,m-1/2}}{q'} \right] \tag{A-5}$$

Some general comments about $T_{i,m}$ are obvious from Eq. (A-5) . If γ_m is integer, a simple polynomial representation for $S_{i,m}$ and $\psi_{i,m-1/2}$ (in q') could permit direct integration of Eq. (A-5) . At best, all integrals could be obtained. At worst, one functional integration, or table lookup of the exponential integral,

$$\int_{q_{i-1/2}}^{q} dq' \frac{e^{\frac{-\sigma q'}{\mu_m}}}{q'} = \left[E_1 (\sigma q_{i-1/2}/\mu_m) - E_1 (\sigma q/\mu_m) \right], \tag{A-6}$$

would be required. A constant or linear representation for $S_{i,m}$ and $\psi_{i,m-1/2}$, is used in slabs, might then extend slab methodology directly to cylinders and spheres (and would recover slab results for $\gamma_m = 0$). The total source term for γ_m integer and $S_{i,m}$, $\psi_{i,m-1/2}$ L^{th} order polynomials is written symbolically,

$$T_{i,m} (q_{i-1/2}, q) = e^{\frac{-\sigma q}{\mu_m}} q^{-\gamma_m} \sum_{\ell=0}^{L} a_{m,\ell} \int_{q_{i-1/2}}^{q} dq' e^{\frac{\sigma q'}{m}} q'^{\gamma_m + \ell} \tag{A-7}$$

with $a_{m,\ell}$ appropriate constants from Eq. (A-5). Unfortunately, except for the S_2 case, γ_m is not generally integer with Gaussian quadrature sets.

If one works on a very fine mesh, such that,

$$\frac{\sigma \Delta q}{\mu_m} << 1, \tag{A-8}$$

in the exponential term, then a Taylor series expansion could be employed to facilitate the reduction,

$$T_{i,m}\,(q_{i-1/2},q) \cong q^{-\gamma_m} \int_{q_{i-1/2}}^{q} dq'\; q'^{\,\gamma_m}$$

$$\left[1 - \frac{\sigma(q-q')}{\mu_m} + \frac{\sigma^2(q-q')^2}{2\mu_m^2} - \ldots\right]$$

$$x\left[\frac{S_{i,m}}{\mu_m} + \frac{\gamma_m \psi_{i,m-1/2}}{q'}\right] \qquad \text{(A-9)}$$

Again, a polynomial representation for $S_{i,m}$, $\psi_{i,m-1/2}$ permits direct evaluation. As reported earlier[5], the expansion to first order in (q-q') was used near the spherical origin to check SNEX.

As both $q_{i-1/2}$, $q \to \infty$, one has $q' \to \infty$ and Eq. (A-5) approaches the expression,

$$T_{i,m}\,(q_{i-1/2},\, q) \cong e^{\frac{-\sigma q}{\mu_m}} \int_{q_{i-1/2}}^{q} dq'\; e^{\frac{\sigma q}{\mu_m}}\; \frac{S_{i,m}}{\mu_m}, \qquad \text{(A-10)}$$

which is the usual slab result. This result is of course required by the transport. Eq. (1) when $q \to \infty$ in cylindrical and spherical geometries.

Apart from the various situations described above, an explicity differenced source term based upon integration of $T_{i,m}$ is difficult excepting for some possibly complicated hybrid respresentations of the external scattering and diamond edge flux sources. However, some simpler approaches can be made.

Removing q'^{γ_m} from inside the integrand and replacing it with $q_i^{\gamma_m}$ on the outside of Eq. (A-5) yields,

$$T_{i,m}\,(q_{i-1/2},q) \quad e^{\frac{-\sigma q}{\mu_m}} \left(\frac{q}{q_i}\right)^{-\gamma_m} \int_{q_{i-1/2}}^{q} dq'\; e^{\frac{\sigma q'}{\mu_m}}$$

$$\left[\frac{S_{i,m}}{\mu_m} + \frac{\gamma_m \psi_{i,m-1/2}}{q'}\right] \qquad \text{(A-11)}$$

which is slab-like, except for the additional term involving $\psi_{i,m-1/2}$. The factor q_i scales the integrand and might be the cell center coordinate, or some other appropriately averaged quantity. If we express $S_{i,m}$ and $\psi_{i,m-1/2}$ as linear functionals in the i^{th} cell,

$$S_{i,m} = S^0_{i,m} + S^1_{i,m}\, q'$$

$$\psi_{i,m-1/2} = \psi^0_{i,m-1/2} + \psi^1_{i,m-1/2}\, q', \qquad \text{(A-12)}$$

with the superscripted quantities constant, and take $q = q_{i+1/2}$, we can write Eq. (A-11) in differenced form,

$$T_{i,m}\,(q_{i-1/2},\, q_{i+1/2}) = \left(\frac{q_{i+1/2}}{q_i}\right)^{-\gamma_m} \left\{ \frac{S^0_{i,m}}{\sigma} \left(1 - e^{\frac{-\sigma\Delta q}{u_m}}\right) \right.$$

$$+ \frac{S^1_{i,m}}{\sigma} \left(q_{i+1/2} - q_{i-1/2}\, e^{\frac{-\sigma\Delta q}{\mu_m}}\right)$$

$$\left. - \frac{S^1_{i,m}\mu_m}{\sigma^2} \left(1 - e^{\frac{-\sigma\Delta q}{\mu_m}}\right) \right.$$

$$+ \psi^{0}_{i,m-1/2} \gamma_m \, e^{\frac{-\sigma q_{i+1/2}}{\mu_m}} \left[E_1(\sigma q_{i-1/2}/\mu_m) - E_1(\sigma q_{i+1/2}\mu/_m) \right]$$

$$+ \frac{\psi^{1}_{i,m-1/2} \mu_m \gamma_m}{\sigma} \left(1 - e^{\frac{-\sigma \Delta q}{\mu_m}} \right) \Bigg\}, \qquad \text{(A-13)}$$

for,

$$\Delta q = q_{i+1/2} - q_{i-1/2}$$

$$2q_i = q_{i+1/2} + q_{i-1/2} \qquad \text{(A-14)}$$

Since, from Eq. (A-4)

$$\psi_{i,m}(q) \sim e^{\frac{-\sigma(q-q_{i\ -1/2})}{\mu_m}} q^{-\gamma_m} \qquad \text{(A-15)}$$

it is tempting to assume a similar exponential representation for $\psi_{i,m-1/2}$ in $T_{i,m}$, (in contrast to the linear expression of Eq. (A-12). We take,

$$\psi_{i,m-1/2}(q') \sim \psi^{3}_{i,m-1/2} \, e^{\frac{-\sigma q'}{\mu_m}} q'^{-\gamma_m} \qquad \text{(A-16)}$$

and $\psi^{3}_{i,m-1/2}$ some appropriately determined constant in the cell. Using the linear source of Eq. (A-12) and the exponential diamond edge flux of Eq. (A-16), another difference scheme is suggested. Inserting Eqs. (26), (A-14) and (30), (A-16) into Eq. (A-5) yields,

$$T_{i,m}(q_{i-1/2},q_{i+1/2}) \cong \left(\frac{q_{i+1/2}}{q_i}\right)^{-\gamma_m} \left\{\frac{S^0_{i,m}}{\sigma}\left(1 - e^{\frac{-\sigma\Delta q}{\mu_m}}\right)\right.$$

$$+ \frac{S^1_{i,m}}{\sigma}\left(q_{i+1/2} - q_{i-1/2}\, e^{\frac{-\sigma\Delta q}{\mu_m}}\right)$$

$$\left. - \frac{S^1_{i,m}\mu_m}{\sigma^2}\left(1 - e^{\frac{-\sigma\Delta q}{\mu_m}}\right)\right\}$$

$$+ \psi^3_{i,m-1/2}\gamma_m\, e^{\frac{-\sigma q_{i+1/2}}{\mu_m}}\, q_{i+1/2}^{-\gamma_m} \ln\left(\frac{q_{i+1/2}}{q_{i-1/2}}\right) \qquad \text{(A-17)}$$

which is simpler than Eq. (A-13) and exhibits the property that the term involving $\psi^3_{i,m-1/2}$ approaches zero as q gets large.

The basic differencing scheme suggested by Eq. (A-2) is exponential with modification of powers of $q^{-\gamma_m}$,

$$\psi_{i,m}(q) = \psi_{i-1/2,m}\, e^{\frac{-\sigma(q-q_{i-1/2})}{\mu_m}} \left(\frac{q}{q_{i-1/2}}\right)^{-\gamma_m}, \qquad \text{(A-18)}$$

while the diamond scheme relates cell centered and edge fluxes only,

$$2\,\psi_{i,m}(q_i) = \psi_{i-1/2,m} + \psi_{i+1/2,m}, \qquad \text{(A-19)}$$

and the linear discontinuous scheme assumes a Lagrangian representation,

$$\psi_{i,m}(q) = \frac{1}{\Delta q}\left[\psi_{i-1/2,m}(q_{i+1/2} - q) + \psi_{i+1/2,m}(q-q_{i-1/2})\right] \qquad \text{(A-20)}$$

Expanding Eq. (A-18) in Taylor series gives the result,

$$\psi_{i,m}(q) \sim \psi_{i-1/2,m} \left\{ 1 - \left(\frac{\sigma}{\mu_m} + \frac{\gamma_m}{q_{i-1/2}} \right) (q - q_{i-1/2}) \right.$$

$$+ \frac{1}{2} \left[\left(\frac{\sigma}{\mu_m} \right)^2 + \frac{\gamma_m(\gamma_m+1)}{q_{i-1/2}^2} \right.$$

$$\left. \left. + \frac{\sigma\gamma_m}{\mu_m q_{i-1/2}} \right] (q - q_{i-1/2})^2 + \cdots \right\} \qquad \text{(A-21)}$$

Substituting the expansion Eq. (A-21) into the diamond Eq. (A-19) yields the identity relationship thru <u>first</u> order. Substituting Eq. (A-21) into Lagrangian Eq. (A-20) yields the identity relationship in <u>zeroth</u> order for $q \neq q_i$ and <u>first</u> order at $q = q_i$.

Printed in the United States of America. Available from
National Technical Information Service
US Department of Commerce
5285 Port Royal Road
Springfield, VA 22161

Microfiche $3.00

001-025	4.00	126-150	7.25	251-275	10.75	376-400	13.00	501-525	15.25
026-050	4.50	151-175	8.00	276-300	11.00	401-425	13.25	526-550	15.50
051-075	5.25	176-200	9.00	301-325	11.75	426-450	14.00	551-575	16.25
076-100	6.00	201-225	9.25	326-350	12.00	451-475	14.50	576-600	16.50
101-125	6.50	226-250	9.50	351-375	12.50	476-500	15.00	601-up	

Note: Add $2.50 for each additional 100-page increment from 601 pages up.